USGS
science for a changing world

Techniques of Water-Resources Investigations
of the United States Geological Survey

Chapter A1

A MODULAR THREE-DIMENSIONAL FINITE-DIFFERENCE GROUND-WATER FLOW MODEL

By Michael G. McDonald and
Arlen W. Harbaugh

This chapter supersedes U.S. Geological Survey Open-File Report 83–875

Book 6

MODELING TECHNIQUES

Narrative for Module BCF1RP

This module reads transmissivity along rows, hydraulic conductivity along rows, storage coefficients, vertical conductance, elevation of top of layer, and elevation of bottom of layer. It also calls SBCF1N to calculate parameters which are constant throughout simulation. It does this in the following order:

1. Call utility module U1DREL to read DELR, DELC, and TRPY which have one value for each column, row, and layer, respectively. TRPY is the ratio of transmissivity along columns to transmissivity along rows for each layer.

2. For each layer, use utility module U2DREL to read the properties of the porous medium. The data requirements for each layer are determined by the layer-type code.

   (a) Find the address of the layer in the three-dimension arrays.

   (b) If the simulation is transient (ISS = 0), read the primary storage coefficient.

   (c) For constant transmissivity layers (LAYCON = 0 or 2), read the transmissivity.

   (d) For variable transmissivity layers (LAYCON = 1 or 3), read hydraulic conductivity and bottom.

   (e) Read vertical-hydraulic conductivity divided by thickness. These values will be multiplied in the program by cell areas to get vertical conductance. For each layer, the vertical conductance to the next lower layer is calculated. Therefore, no vertical conductance is calculated for the lowest layer in the mesh.

   (f) If the simulation is transient and the layer type is two or three, read the secondary storage coefficient (specific yield).

   (g) Read the top elevation if the layer type is two or three.

3. Call SBCF1N to calculate conductance and storage terms which are constant during the simulation and check to see that branch conductances agree with boundaries specified in the IBOUND array.

4. RETURN.

Flow Chart for Module BCF1RP

DELR is the grid spacing in the row direction.

DELC is the grid spacing in the column direction.

TRPY is the ratio of transmissivity in the column direction to transmissivity in the row direction.

LAYCON is a layer-type code (one for each layer).

- 0 - confined
- 1 - unconfined
- 2 - confined/unconfined but transmissivity is constant
- 3 - confined/unconfined

Secondary Storage coefficient is relevant only for convertible layers (LAYCON = 2 or 3); then it is equal to specific yield.



```
      SUBROUTINE BCF1RP(IBOUND,HNEW,SC1,HY,CR,CC,CV,DELR,DELC,
     1        BOT,TOP,SC2,TRPY,IN,ISS,NCOL,NROW,NLAY,NODES,IOUT)
C
C-----VERSION 1636 15MAY1987 BCF1RP
C
C     ******************************************************************
C     READ AND INITIALIZE DATA FOR BLOCK-CENTERED FLOW PACKAGE
C     ******************************************************************
C
C        SPECIFICATIONS:
C     ------------------------------------------------------------------
      CHARACTER*4 ANAME
      DOUBLE PRECISION HNEW
C
      DIMENSION HNEW(NODES),SC1(NODES),HY(NODES),CR(NODES),CC(NODES),
     1          CV(NODES),ANAME(6,10),DELR(NCOL),DELC(NROW),BOT(NODES),
     1          TOP(NODES),SC2(NODES),TRPY(NLAY),IBOUND(NODES)
C
      COMMON /FLWCOM/LAYCON(80)
C
      DATA ANAME(1,1),ANAME(2,1),ANAME(3,1),ANAME(4,1),ANAME(5,1),
     1  ANAME(6,1) /'    ','PRIM','ARY ','STOR','AGE ','COEF'/
      DATA ANAME(1,2),ANAME(2,2),ANAME(3,2),ANAME(4,2),ANAME(5,2),
     1  ANAME(6,2) /'    ','TRAN','SMIS','. AL','ONG ','ROWS'/
      DATA ANAME(1,3),ANAME(2,3),ANAME(3,3),ANAME(4,3),ANAME(5,3),
     1  ANAME(6,3) /'   H','YD. ','COND','. AL','ONG ','ROWS'/
      DATA ANAME(1,4),ANAME(2,4),ANAME(3,4),ANAME(4,4),ANAME(5,4),
     1  ANAME(6,4) /'VERT',' HYD',' CON','D /T','HICK','NESS'/
      DATA ANAME(1,5),ANAME(2,5),ANAME(3,5),ANAME(4,5),ANAME(5,5),
     1  ANAME(6,5) /'    ','    ','    ','    ','  BO','TTOM'/
      DATA ANAME(1,6),ANAME(2,6),ANAME(3,6),ANAME(4,6),ANAME(5,6),
     1  ANAME(6,6) /'    ','    ','    ','    ','    ',' TOP'/
      DATA ANAME(1,7),ANAME(2,7),ANAME(3,7),ANAME(4,7),ANAME(5,7),
     1  ANAME(6,7) /'  SE','COND','ARY ','STOR','AGE ','COEF'/
      DATA ANAME(1,8),ANAME(2,8),ANAME(3,8),ANAME(4,8),ANAME(5,8),
     1  ANAME(6,8) /'COLU','MN T','O RO','W AN','ISOT','ROPY'/
      DATA ANAME(1,9),ANAME(2,9),ANAME(3,9),ANAME(4,9),ANAME(5,9),
     1  ANAME(6,9) /'    ','    ','    ','    ','    ','DELR'/
      DATA ANAME(1,10),ANAME(2,10),ANAME(3,10),ANAME(4,10),ANAME(5,10),
     1  ANAME(6,10) /'    ','    ','    ','    ','    ','DELC'/
C     ------------------------------------------------------------------
C
C1------CALCULATE NUMBER OF NODES IN A LAYER AND READ TRPY,DELR,DELC
      NIJ=NCOL*NROW
C
      CALL U1DREL(TRPY,ANAME(1,8),NLAY,IN,IOUT)
      CALL U1DREL(DELR,ANAME(1,9),NCOL,IN,IOUT)
      CALL U1DREL(DELC,ANAME(1,10),NROW,IN,IOUT)
```

```
C
C2------READ ALL PARAMETERS FOR EACH LAYER
      KT=0
      KB=0
      DO 200 K=1,NLAY
      KK=K
C
C2A-----FIND ADDRESS OF EACH LAYER IN THREE DIMENSION ARRAYS.
      IF(LAYCON(K).EQ.1 .OR. LAYCON(K).EQ.3) KB=KB+1
      IF(LAYCON(K).EQ.2 .OR. LAYCON(K).EQ.3) KT=KT+1
      LOC=1+(K-1)*NIJ
      LOCB=1+(KB-1)*NIJ
      LOCT=1+(KT-1)*NIJ
C
C2B-----READ PRIMARY STORAGE COEFFICIENT INTO ARRAY SC1 IF TRANSIENT
      IF(ISS.EQ.0)CALL U2DREL(SC1(LOC),ANAME(1,1),NROW,NCOL,KK,IN,IOUT)
C
C2C-----READ TRANSMISSIVITY INTO ARRAY CC IF LAYER TYPE IS 0 OR 2
      IF(LAYCON(K).EQ.3 .OR. LAYCON(K).EQ.1) GO TO 100
      CALL U2DREL(CC(LOC),ANAME(1,2),NROW,NCOL,KK,IN,IOUT)
      GO TO 110
C
C2D-----READ HYDRAULIC CONDUCTIVITY(HY) AND BOTTOM ELEVATION(BOT)
C2D-----IF LAYER TYPE IS 1 OR 3
  100 CALL U2DREL(HY(LOCB),ANAME(1,3),NROW,NCOL,KK,IN,IOUT)
      CALL U2DREL(BOT(LOCB),ANAME(1,5),NROW,NCOL,KK,IN,IOUT)
C
C2E-----READ VERTICAL HYCOND/THICK INTO ARRAY CV IF NOT BOTTOM LAYER
C2E----- READ AS HYCOND/THICKNESS -- CONVERTED TO CONDUCTANCE LATER
  110 IF(K.EQ.NLAY) GO TO 120
      CALL U2DREL(CV(LOC),ANAME(1,4),NROW,NCOL,KK,IN,IOUT)
C
C2F-----READ SECONDARY STORAGE COEFFICIENT INTO ARRAY SC2 IF TRANSIENT
C2F-----AND LAYER TYPE IS 2 OR 3
  120 IF(LAYCON(K).NE.3 .AND. LAYCON(K).NE.2) GO TO 200
      IF(ISS.EQ.0)CALL U2DREL(SC2(LOCT),ANAME(1,7),NROW,NCOL,KK,IN,IOUT)
C
C2G-----READ TOP ELEVATION(TOP) IF LAYER TYPE IS 2 OR 3
      CALL U2DREL(TOP(LOCT),ANAME(1,6),NROW,NCOL,KK,IN,IOUT)
  200 CONTINUE
C
C3------PREPARE AND CHECK BCF DATA
      CALL SBCF1N(HNEW,IBOUND,SC1,SC2,CR,CC,CV,HY,TRPY,DELR,DELC,ISS,
     1          NCOL,NROW,NLAY,IOUT)
C
C4------RETURN
      RETURN
      END
```

List of Variables for Module BCF1RP

Variable	Range	Definition
ANAME	Module	Label for printout of input array.
BOT	Package	DIMENSION (NCOL,NROW,NBOT), Elevation of the bottom of each layer. (NBOT is the number of layers for which LAYCON = 1 or 3.)
CC	Global	DIMENSION (NCOL,NROW,NLAY), Conductance in the column direction. CC(J,I,K) contains conductance between nodes (J,I,K) and (J,I+1,K), This array is used to temporarily hold transmissivity.
CR	Global	DIMENSION (NCOL,NROW,NLAY), Conductance in the row direction. CR(J,I,K) contains conductance between nodes (J,I,K) and (J+1,I,K).
CV	Global	DIMENSION (NCOL,NROW,NLAY-1), Conductance in the vertical direction. CV(J,I,K) contains conductance between nodes (J,I,K) and (J,I,K+1). This array is used to temporarily to hold Vcont.
DELC	Global	DIMENSION (NROW), Cell dimension in the column direction. DELC(I) contains width of row I.
DELR	Global	DIMENSION (NCOL), Cell dimension in the row direction. DELR(J) contains the width of column J.
HNEW	Global	DIMENSION (NCOL,NROW,NLAY), Most recent estimate of head in each cell. HNEW changes at each iteration.
HY	Package	DIMENSION (NCOL,NROW,NBOT), Hydraulic conductivity of a cell. (NBOT is the number of layers where LAYCON = 1 or 3.)
IBOUND	Global	DIMENSION (NCOL,NROW,NLAY), Status of each cell. < 0, constant-head cell = 0, inactive cell > 0, variable-head cell
IN	Package	Primary unit number from which input for this package will be read.
IOUT	Global	Primary unit number for all printed output. IOUT = 6.
ISS	Package	Flag. = 0, simulation is transient. ≠ 0, simulation is steady state.
K	Module	Index for layers.
KB	Module	Counter for the number of layers for which the bottom elevation is needed (LAYCON = 1 or 3).
KK	Module	Temporary variable set equal to K. KK is used as an actual argument in subroutine calls to avoid using the DO loop variable K as an argument, which causes problems with some compilers.
KT	Module	Counter for the number of layers for which the top elevation is needed (LAYCON = 2 or 3).
LAYCON	Package	DIMENSION (80) Layer type code: 0 - Layer strictly confined. 1 - Layer strictly unconfined. 2 - Layer confined/unconfined (transmissivity is constant). 3 - Layer confined/unconfined (transmissivity

List of Variables for Module BCF1RP (Continued)

Variable	Range	Definition
LOC	Module	Pointer to parts of the conductance arrays corresponding to particular layers.
LOCB	Module	Pointer to parts of the BOT and HY arrays corresponding to particular layers.
LOCT	Module	Pointer to parts of the TOP and SC1 arrays corresponding to particular layers.
NCOL	Global	Number of columns in the grid.
NIJ	Module	Number of cells in a layer.
NLAY	Global	Number of layers in the grid.
NODES	Global	Number of cells (nodes) in the finite-difference grid.
NROW	Global	Number of rows in the grid.
†SC1	Package	DIMENSION (NCOL,NROW,NLAY), Primary storage capacity of each cell (S*DELC*DELR).
†SC2	Package	DIMENSION (NCOL,NROW,NTOP), Secondary storage capacity of each cell in the grid. (NTOP is the number of layers for which LAYCON = 2 or 3.)
TOP	Package	DIMENSION (NCOL,NROW,NTOP), Elevation of the top of the layers. (NTOP is the number of layers for which LAYCON = 2 or 3.)
TRPY	Package	DIMENSION (NLAY), Ratio of transmissivity in the column direction to transmissivity in the row direction.

†Initially, storage coefficient values are read into these arrays; these values are multiplied by cell areas in submodule SBCF1N to yield storage capacities.

Narrative for Module BCF1FM

This module calculates branch conductances which are not constant throughout the simulation, adds storage terms to the accumulators in which HCOF and RHS are formed, and adds terms to RHS and HCOF which correct for overestimation of flow down into partially saturated cells.

1. For each layer in which transmissivity varies with head (LAYCON = 1 or 3), call submodule SBCF1H to calculate branch conductance.

2. If the simulation is transient, calculate storage terms (STEPS 3-5) for each layer. If the simulation is steady state, GO TO STEP 6.

†3. Determine if there is one storage factor or two.

†4. If there is only one storage factor (LAYCON = 0 or 1), use it to calculate storage terms and add them to the right hand side (RHS) and the h-coefficient (HCOF).

†5. If there are two storage factors, then, using head at the beginning of the time step (HOLD), determine the storage factor at the beginning of the time step (SOLD) and use the latest estimate of head at the end of the time step (HNEW) to determine the storage factor at the end of the time step (SNEW). Use SOLD and SNEW to calculate the storage terms to add to RHS and HCOF.

6. For each layer, determine if correction terms are needed for flow down into a partially saturated layer (STEPS 7-8).

7. If the layer is partially saturated and there is flow from above, calculate correction terms and add to RHS and HCOF.

8. If this is not the bottom layer and the layer below is partially saturated, calculate the correction terms and add to RHS and HCOF.

9. RETURN.

†The term storage factor, as used in Subroutine BCF1FM, refers to storage capacity divided by time step length. SOLD is thus equivalent here to $SCA/(t_m-t_{m-1})$, in the notation of equations (61) and (62), while SNEW is equivalent to $SCB/(t_m-t_{m-1})$.

Flow Chart for Module BCF1FM

LAYCON is a layer-type code
(one for each layer).

0 - confined
1 - unconfined
2 - confined/unconfined
but transmissivity
is constant
3 - confined/unconfined



```
      SUBROUTINE BCF1FM(HCOF,RHS,HOLD,SC1,HNEW,IBOUND,CR,CC,CV,HY,TRPY,
     1                  BOT,TOP,SC2,DELR,DELC,DELT,ISS,KITER,KSTP,KPER,
     2                  NCOL,NROW,NLAY,IOUT)
C-----VERSION 1640 15MAY1987 BCF1FM
C
C     ******************************************************************
C     ADD LEAKAGE CORRECTION AND STORAGE TO HCOF AND RHS, AND CALCULATE
C     CONDUCTANCE AS REQUIRED
C     ******************************************************************
C
C        SPECIFICATIONS:
C     ------------------------------------------------------------------
      DOUBLE PRECISION HNEW
C
      DIMENSION HCOF(NCOL,NROW,NLAY),RHS(NCOL,NROW,NLAY),
     1    HOLD(NCOL,NROW,NLAY),SC1(NCOL,NROW,NLAY),HNEW(NCOL,NROW,NLAY),
     2    IBOUND(NCOL,NROW,NLAY),CR(NCOL,NROW,NLAY),
     3    CC(NCOL,NROW,NLAY),CV(NCOL,NROW,NLAY),HY(NCOL,NROW,NLAY),
     4    TRPY(NLAY),BOT(NCOL,NROW,NLAY),TOP(NCOL,NROW,NLAY),DELR(NCOL),
     5    DELC(NROW),SC2(NCOL,NROW,NLAY)
C
      COMMON /FLWCOM/LAYCON(80)
C     ------------------------------------------------------------------
      KB=0
      KT=0
C
C1------FOR EACH LAYER: IF T VARIES CALCULATE HORIZONTAL CONDUCTANCES
      DO 100 K=1,NLAY
      KK=K
      IF(LAYCON(K).EQ.3 .OR. LAYCON(K).EQ.2) KT=KT+1
C
C1A-----IF LAYER TYPE IS NOT 1 OR 3 THEN SKIP THIS LAYER.
      IF(LAYCON(K).NE.3 .AND. LAYCON(K).NE.1) GO TO 100
      KB=KB+1
C
C1B-----FOR LAYER TYPES 1 & 3 CALL SBCFH1 TO CALCULATE
C1B-----HORIZONTAL CONDUCTANCES.
      CALL SBCF1H(HNEW,IBOUND,CR,CC,CV,HY,TRPY,DELR,DELC,BOT,TOP,
     1         KK,KB,KT,KITER,KSTP,KPER,NCOL,NROW,NLAY,IOUT)
  100 CONTINUE
C
C2------IF THE SIMULATION IS TRANSIENT ADD STORAGE TO HCOF AND RHS
      IF(ISS.NE.0) GO TO 201
      TLED=1./DELT
      KT=0
      DO 200 K=1,NLAY
C
C3------SEE IF THIS LAYER IS CONVERTIBLE OR NON-CONVERTIBLE.
      IF(LAYCON(K).EQ.3 .OR. LAYCON(K).EQ.2) GO TO 150
C4------NON-CONVERTIBLE LAYER, SO USE PRIMARY STORAGE
      DO 140 I=1,NROW
      DO 140 J=1,NCOL
      IF(IBOUND(J,I,K).LE.0) GO TO 140
      RHO=SC1(J,I,K)*TLED
      HCOF(J,I,K)=HCOF(J,I,K)-RHO
      RHS(J,I,K)=RHS(J,I,K)-RHO*HOLD(J,I,K)
  140 CONTINUE
      GO TO 200
C
C5------A CONVERTIBLE LAYER, SO CHECK OLD AND NEW HEADS TO DETERMINE
C5------WHEN TO USE PRIMARY AND SECONDARY STORAGE
  150 KT=KT+1
      DO 180 I=1,NROW
      DO 180 J=1,NCOL
C
C5A-----IF THE CELL IS EXTERNAL THEN SKIP IT.
      IF(IBOUND(J,I,K).LE.0) GO TO 180
      TP=TOP(J,I,KT)
      RHO2=SC2(J,I,KT)*TLED
```

```
      RHO1=SC1(J,I,K)*TLED
C
C5B-----FIND STORAGE FACTOR AT START OF TIME STEP.
      SOLD=RHO2
      IF(HOLD(J,I,K).GT.TP) SOLD=RHO1
C
C5C-----FIND STORAGE FACTOR AT END OF TIME STEP.
      HTMP=HNEW(J,I,K)
      SNEW=RHO2
      IF(HTMP.GT.TP) SNEW=RHO1
C
C5D-----ADD STORAGE TERMS TO RHS AND HCOF.
      HCOF(J,I,K)=HCOF(J,I,K)-SNEW
      RHS(J,I,K)=RHS(J,I,K) - SOLD*(HOLD(J,I,K)-TP) - SNEW*TP
C
  180 CONTINUE
C
  200 CONTINUE
C
C6------FOR EACH LAYER DETERMINE IF CORRECTION TERMS ARE NEEDED FOR
C6------FLOW DOWN INTO PARTIALLY SATURATED LAYERS.
  201 KT=0
      DO 300 K=1,NLAY
C
C7------SEE IF CORRECTION IS NEEDED FOR LEAKAGE FROM ABOVE.
      IF(LAYCON(K).NE.3 .AND. LAYCON(K).NE.2) GO TO 250
      KT=KT+1
      IF(K.EQ.1) GO TO 250
C
C7A-----FOR EACH CELL MAKE THE CORRECTION IF NEEDED.
      DO 220 I=1,NROW
      DO 220 J=1,NCOL
C
C7B-----IF THE CELL IS EXTERNAL(IBOUND<=0) THEN SKIP IT.
      IF(IBOUND(J,I,K).LE.0) GO TO 220
      HTMP=HNEW(J,I,K)
C
C7C-----IF HEAD IS ABOVE TOP THEN CORRECTION NOT NEEDED
      IF(HTMP.GE.TOP(J,I,KT)) GO TO 220
C
C7D-----WITH HEAD BELOW TOP ADD CORRECTION TERMS TO RHS AND HCOF.
      RHS(J,I,K)=RHS(J,I,K) + CV(J,I,K-1)*TOP(J,I,KT)
      HCOF(J,I,K)=HCOF(J,I,K) + CV(J,I,K-1)
  220 CONTINUE
C
C8------SEE IF THIS LAYER MAY NEED CORRECTION FOR LEAKAGE TO BELOW.
  250 IF(K.EQ.NLAY) GO TO 300
      IF(LAYCON(K+1).NE.3 .AND. LAYCON(K+1).NE.2) GO TO 300
      KTT=KT+1
C
C8A-----FOR EACH CELL MAKE THE CORRECTION IF NEEDED.
      DO 280 I=1,NROW
      DO 280 J=1,NCOL
C
C8B-----IF CELL IS EXTERNAL (IBOUND<=0) THEN SKIP IT.
      IF(IBOUND(J,I,K).LE.0) GO TO 280
C
C8C-----IF HEAD IN THE LOWER CELL IS LESS THAN TOP ADD CORRECTION
C8C-----TERM TO RHS.
      HTMP=HNEW(J,I,K+1)
      IF(HTMP.LT.TOP(J,I,KTT)) RHS(J,I,K)=RHS(J,I,K)
     1                         - CV(J,I,K)*(TOP(J,I,KTT)-HTMP)
  280 CONTINUE
  300 CONTINUE
C
C9------RETURN
      RETURN
      END
```

List of Variables for Module BCF1FM

Variable	Range	Definition
BOT	Package	DIMENSION (NCOL,NROW,NBOT), Elevation of bottom of each layer. (NBOT is the number of layers for which LAYCON = 1 or 3.)
CC	Global	DIMENSION (NCOL,NROW,NLAY), Conductance in the column direction. CC(J,I,K) contains conductance between nodes (J,I,K) and (J,I+1,K).
CR	Global	DIMENSION (NCOL,NROW,NLAY), Conductance in the row direction. CR(J,I,K) contains conductance between nodes (J,I,K) and (J+1,I,K).
CV	Global	DIMENSION (NCOL,NROW,NLAY-1), Conductance in the vertical direction. CV(J,I,K) contains conductance between nodes (J,I,K) and (J,I,K+1).
DELC	Global	DIMENSION (NROW), Cell dimension in the column direction. DELC(I) contains the width of row I.
DELR	Global	DIMENSION (NCOL), Cell dimension in the row direction. DELR(J) contains the width of column J.
DELT	Global	Length of the current time step.
HCOF	Global	DIMENSION (NCOL,NROW,NLAY), Coefficient of head in cell (J,I,K) in the finite-difference equation.
HNEW	Global	DIMENSION (NCOL,NROW,NLAY), Most recent estimate of head in each cell. HNEW changes at each iteration.
HOLD	Global	DIMENSION (NCOL,NROW,NLAY), Head at the start of the current time step.
HTMP	Module	Temporary single precision HNEW(J,I,K).
HY	Package	DIMENSION (NCOL,NROW,NBOT), Hydraulic conductivity of a cell. (NBOT is the number of layers where LAYCON = 1 or 3.)
I	Module	Index for rows.
IBOUND	Global	DIMENSION (NCOL,NROW,NLAY), Status of each cell. < 0, constant-head cell = 0, inactive cell > 0, variable-head cell
IOUT	Global	Primary unit number for all printed output. IOUT = 6.
ISS	Package	Flag. = 0, simulation is transient. ≠ 0, simulation is steady state.
J	Module	Index for columns.
K	Module	Index for layers.
KB	Module	Counter for layers for which bottom elevation is needed.
KITER	Global	Iteration counter. Reset at the start of each time step.
KK	Module	Temporary variable set equal to K. KK is used as an actual argument in subroutine calls to avoid using the DO loop variable K as an argument, which causes problems with some compilers.
KPER	Global	Stress period counter.
KSTP	Global	Time step counter. Reset at the start of each stress period.
KT	Module	Counter for layers for which top elevation is needed.
KTT	Module	Pointer to TOP array of layer immediately below layer K.

List of Variables for Module BCF1FM (Continued)

Variable	Range	Definition
LAYCON	Package	DIMENSION (80) Layer type code: 0 - Layer strictly confined. 1 - Layer strictly unconfined. 2 - Layer confined/unconfined (transmissivity is constant). 3 - Layer confined/unconfined (transmissivity varies).
NCOL	Global	Number of columns in the grid.
NLAY	Global	Number of layers in the grid.
NROW	Global	Number of rows in the grid.
RHO	Module	Storage coefficient for strictly confined or strictly unconfined layers.
†RHO1	Module	Confined storage factor for convertible layers.
†RHO2	Module	Unconfined storage factor for convertible layers.
RHS	Global	DIMENSION (NCOL,NROW,NLAY), Right hand side of finite-difference equation. RHS is an accumulation of terms from several different packages.
SC1	Package	DIMENSION (NCOL,NROW,NLAY), Primary storage capacity of each cell (S*DELC*DELR).
SC2	Package	DIMENSION (NCOL,NROW,NTOP), Secondary storage capacity of each cell in the grid. (NTOP is the number of layers for which LAYCON = 2 or 3.)
†SNEW	Module	Storage factor at the end of the time step for convertible layers.
†SOLD	Module	Storage factor at the start of the time step for convertible layers.
TLED	Module	1/DELT.
TOP	Package	DIMENSION (NCOL,NROW,NTOP), Elevation of top of layers. (NTOP is the number of layers for which LAYCON = 2 or 3.)
TP	Module	Temporary variable for TOP(J,I,K).
TRPY	Package	DIMENSION (NLAY), Ratio of transmissivity in the column direction to transmissivity in the row direction.

†Storage factor, as used in Subroutine BCF1FM, refers to storage capacity divided by time step length.

Narrative for Module BCF1BD

Module BCF1BD calculates flow rates within the porous medium for use in the overall volumetric budget and calculates cell-by-cell flow terms for recording on disk. Flow rates to constant heads and from storage are accumulated and passed to the module BAS1OT for inclusion in the budget. They are accumulated by sign so that flow into constant-head cells is separate from flow out of constant-head cells, and flow into storage is separate from flow out of storage. Flow rates to constant-head cells and from storage as well as flow across cell boundaries can be recorded on a cell-by-cell basis for use by other programs.

Flow from storage is calculated inside BCF1BD. Flow to constant-head cells and across cell boundaries is calculated in submodules SBCF1F and SBCF1B, respectively.

Module BCF1BD performs its tasks in the following order:

1. Clear the fields STOIN and STOUT in which flow out of and into storage, respectively, are accumulated.

2. If the user has specified that cell-by-cell flow terms should be recorded this time step (ICBCFL ≠0) and has specified a unit number (IBCFCB) for cell-by-cell flow terms for the BCF Package, set the cell-by-cell flag (IBD).

3. If this is steady-state simulation, skip all of the calculations for flow from storage.

4. If cell-by-cell flow terms are to be saved (i.e., if IBD was set in STEP 2), clear the buffer (BUFF) in which they will be accumulated prior to printing.

5. For each cell in the grid, calculate flow from storage and move to accumulator (STEPS 6 AND 7).

6. Calculate flow from storage in the cell.

7. If the cell-by-cell rates are being recorded, store flow rate from storage in the buffer. Depending on the sign, add the flow from storage to the accumulators STOIN or STOUT.

8. If the cell-by-cell flag (IBD) is set, record the contents of the buffer.

9. Store the accumulated rates and volumes of flow from storage in table VBVL for inclusion in the overall volumetric budget. Store an appropriate label in the corresponding location in the table VBNM.

10. Call submodule SBCF1F to calculate flow from constant-head cells.

11. If the cell-by-cell flag (IBD) is set, call submodule SBCF1B to calculate and record the flow across cell boundaries.

12. RETURN.

Flow Chart for Module BCF1BD

STOIN is an accumulator for flow terms having a positive sign (flow from storage into the flow system) for inclusion in the volumetric budget.

STOUT is an accumulator for flow terms having a negative sign (flow into storage and out of the flow system) for inclusion in the volumetric budget.

IBD is a flag which indicates that for this time step, BCF cell-by-cell flow terms should be recorded.

BUFF is a buffer where flow terms are gathered prior to recording them.

VBVL is a table of budget entries calculated by component-of-flow packages for use in calculating the volumetric budget.

VBNM is a table of labels for budget terms.



```
      SUBROUTINE BCF1BD(VBNM,VBVL,MSUM,HNEW,IBOUND,HOLD,SC1,CR,CC,CV,
     1   TOP,SC2,DELT,ISS,NCOL,NROW,NLAY,KSTP,KPER,IBCFCB,
     2   ICBCFL,BUFF,IOUT)
C-----VERSION 1546 12MAY1987 BCF1BD
C
C     ******************************************************************
C     COMPUTE BUDGET FLOW TERMS FOR BCF -- STORAGE, CONSTANT HEAD, AND
C     FLOW ACROSS CELL WALLS
C     ******************************************************************
C
C     SPECIFICATIONS:
C     ------------------------------------------------------------------
      CHARACTER*4 VBNM,TEXT
      DOUBLE PRECISION HNEW
C
      DIMENSION HNEW(NCOL,NROW,NLAY), IBOUND(NCOL,NROW,NLAY),
     1   HOLD(NCOL,NROW,NLAY), SC1(NCOL,NROW,NLAY),
     2   CR(NCOL,NROW,NLAY), CC(NCOL,NROW,NLAY),
     3   CV(NCOL,NROW,NLAY), VBNM(4,20), VBVL(4,20),
     4   SC2(NCOL,NROW,NLAY),
     5   TOP(NCOL,NROW,NLAY),BUFF(NCOL,NROW,NLAY)
C
      COMMON /FLWCOM/LAYCON(80)
C
      DIMENSION TEXT(4)
C
      DATA TEXT(1),TEXT(2),TEXT(3),TEXT(4) /'    ','    ',' STO','RAGE'/
C     ------------------------------------------------------------------
C
C1------INITIALIZE BUDGET ACCUMULATORS
      STOIN=0.
      STOUT=0.
C
C2------IF CELL-BY-CELL FLOWS ARE NEEDED THEN SET FLAG IBD.
      IBD=0
      IF(ICBCFL.NE.0 .AND. IBCFCB.GT.0) IBD=1
C
C3------IF STEADY STATE THEN SKIP ALL STORAGE CALCULATIONS
      IF(ISS.NE.0) GO TO 305
C
C4------IF CELL-BY-CELL FLOWS ARE NEEDED (IBD IS SET) CLEAR BUFFER
      IF(IBD.EQ.0) GO TO 220
      DO 210 K=1,NLAY
      DO 210 I=1,NROW
      DO 210 J=1,NCOL
      BUFF(J,I,K)=0.
  210 CONTINUE
C
C5------RUN THROUGH EVERY CELL IN THE GRID
  220 KT=0
      DO 300 K=1,NLAY
      LC=LAYCON(K)
      IF(LC.EQ.3 .OR. LC.EQ.2) KT=KT+1
      DO 300 I=1,NROW
      DO 300 J=1,NCOL
C
C6------CALCULATE FLOW FROM STORAGE (VARIABLE HEAD CELLS ONLY)
```

```
      IF(IBOUND(J,I,K).LE.0) GO TO 300
      HSING=HNEW(J,I,K)
C
C6A----CHECK LAYER TYPE TO SEE IF ONE STORAGE CAPACITY OR TWO
      IF(LC.NE.3 .AND. LC.NE.2) GO TO 285
C
C6B----TWO STORAGE CAPACITIES
      TP=TOP(J,I,KT)
      SYA=SC2(J,I,KT)
      SCFA=SC1(J,I,K)
      SOLD=SYA
      IF(HOLD(J,I,K).GT.TP) SOLD=SCFA
      SNEW=SYA
      IF(HSING.GT.TP) SNEW=SCFA
      STRG=SOLD*(HOLD(J,I,K)-TP) + SNEW*TP - SNEW*HSING
      GO TO 288
C
C6C----ONE STORAGE CAPACITY
  285 SC=SC1(J,I,K)
      STRG=SC*HOLD(J,I,K) - SC*HSING
C
C7-----STORE CELL-BY-CELL FLOW IN BUFFER AND ADD TO ACCUMULATORS
  288 IF(IBD.EQ.1) BUFF(J,I,K)=STRG/DELT
      IF(STRG) 292,300,294
  292 STOUT=STOUT-STRG
      GO TO 300
  294 STOIN=STOIN+STRG
C
  300 CONTINUE
C
C8-----IF IBD FLAG IS SET RECORD THE CONTENTS OF THE BUFFER
      IF(IBD.EQ.1) CALL UBUDSV(KSTP,KPER,TEXT,
     1                         IBCFCB,BUFF,NCOL,NROW,NLAY,IOUT)
C
C9------ADD TOTAL RATES AND VOLUMES TO VBVL & PUT TITLES IN VBNM
  305 VBVL(1,MSUM)=VBVL(1,MSUM)+STOIN
      VBVL(2,MSUM)=VBVL(2,MSUM)+STOUT
      VBVL(3,MSUM)=STOIN/DELT
      VBVL(4,MSUM)=STOUT/DELT
      VBNM(1,MSUM)=TEXT(1)
      VBNM(2,MSUM)=TEXT(2)
      VBNM(3,MSUM)=TEXT(3)
      VBNM(4,MSUM)=TEXT(4)
      MSUM=MSUM+1
C
C10-----CALCULATE FLOW FROM CONSTANT HEAD NODES
      CALL SBCF1F(VBNM,VBVL,MSUM,HNEW,IBOUND,CR,CC,CV,TOP,DELT,
     1        NCOL,NROW,NLAY,KSTP,KPER,IBD,IBCFCB,ICBCFL,BUFF,IOUT)
C
C11-----CALCULATE AND SAVE FLOW ACROSS CELL BOUNDARIES IF C-B-C
C11-----FLOW TERMS ARE REQUESTED.
      IF(IBD.NE.0) CALL SBCF1B(HNEW,IBOUND,CR,CC,CV,TOP,NCOL,NROW,NLAY,
     1         KSTP,KPER,IBCFCB,BUFF,IOUT)
C
C12----RETURN
      RETURN
      END
```

List of Variables for Module BCF1BD

Variable	Range	Definition
BUFF	Global	DIMENSION (NCOL,NROW,NLAY), Buffer used to accumulate information before printing or recording it.
CC	Global	DIMENSION (NCOL,NROW,NLAY), Conductance in the column direction. CC(J,I,K) contains the conductance between nodes (J,I,K) and (J,I+1,K).
CR	Global	DIMENSION (NCOL,NROW,NLAY), Conductance in the row direction. CR(J,I,K) contains conductance between nodes (J,I,K) and (J+1,I,K).
CV	Global	DIMENSION (NCOL,NROW,NLAY-1), Conductance in the vertical direction. CV(J,I,K) contains conductance between nodes (J,I,K) and (J,I,K+1).
DELT	Global	Length of the current time step.
HNEW	Global	DIMENSION (NCOL,NROW,NLAY), Most recent estimate of head in each cell. HNEW changes at each iteration.
HOLD	Global	DIMENSION (NCOL,NROW,NLAY), Head at the start of the current time step.
HSING	Module	Temporary label for element of HNEW.
I	Module	Index for rows.
IBCFCB	Package	Flag and a unit number. > 0, unit number on which the cell-by-cell flow terms will be recorded whenever ICBCFL is set. = 0, cell-by-cell flow terms will not be printed or recorded. < 0, flow from each constant-head cell will be printed whenever ICBCFL is set.
IBD	Package	Flag. = 0, cell-by-cell flow terms for this package will not be recorded. ≠ 0, cell-by-cell flow terms for this package will be recorded.
IBOUND	Global	DIMENSION (NCOL,NROW,NLAY), Status of each cell. < 0, constant-head cell = 0, inactive cell > 0, variable-head cell
ICBCFL	Global	Flag. = 0, cell-by-cell flow terms will not be recorded or printed for the current time step. ≠ 0, cell-by-cell flow terms (flow to constant heads) will be either printed or recorded (depending on IBCFCB) for the current time step.
IOUT	Global	Primary unit number for all printed output. IOUT = 6.
ISS	Package	Flag. = 0, simulation is transient. ≠ 0, simulation is steady state.
J	Module	Index for columns.
K	Module	Index for layers.
KPER	Global	Stress period counter.

List of Variables for Module BCF1BD (continued)

Variable	Range	Definition
KSTP	Global	Time step counter. Reset at the start of each stress period.
KT	Module	Index for top of layers (also used for secondary storage terms).
LAYCON	Package	DIMENSION (80) Layer type code: 0 - Layer strictly confined. 1 - Layer strictly unconfined. 2 - Layer confined/unconfined (transmissivity is constant). 3 - Layer confined/unconfined (transmissivity varies).
LC	Module	Temporary name for LAYCON(K).
MSUM	Global	Counter for budget entries and labels in VBVL and VBNM.
NCOL	Global	Number of columns in the grid.
NLAY	Global	Number of layers in the grid.
NROW	Global	Number of rows in the grid.
SC	Module	Temporary name for the storage capacity.
SCFA	Module	Temporary name for the primary storage capacity.
SC1	Package	DIMENSION (NCOL, NROW, NLAY), Primary storage capacity of each cell (S*DELC*DELR).
SC2	Package	DIMENSION (NCOL, NROW, NTOP), Secondary storage capacity of each cell in the grid. (NTOP is the number of layers for which LAYCON = 2 or 3.)
SNEW*	Module	Storage capacity at the end of the time step.
SOLD*	Module	Storage capacity at the start of the time step.
STOIN	Module	Sum of decreases in storage from individual cells.
STOUT	Module	Sum of increases in storage for individual cells.
STRG	Module	Volume of flow into or out of storage in a single cell.
SYA	Module	Temporary name for the secondary storage capacity.
TEXT	Module	Labels recorded along with the cell-by-cell flow terms.
TOP	Package	DIMENSION (NCOL, NROW, NTOP), Elevation of top of layers. (NTOP is the number of layers for which LAYCON = 2 or 3.)
TP	Module	Temporary label for TOP(J,I,K).
VBNM	Global	DIMENSION(4,20), Labels for entries in the volumetric budget.
VBVL	Global	DIMENSION(4,20), Entries for the volumetric budget. For flow component N, the values in VBVL are: (1,N), Rate for the current time step into the flow field. (2,N), Rate for the current time step out of the flow field. (3,N), Volume into the flow field during simulation. (4,N), Volume out of the flow field during simulation.

*Note that the variables SOLD and SNEW have different meanings in this subroutine than in BCF1FM.

Narrative for Module SBCF1N

This module insures that the transmissive properties of each cell agree with the codes specified in the boundary array (IBOUND) and calculates (1) horizontal-branch conductance in layers where transmissivity is constant, (2) vertical-branch conductance, and (3) storage capacity.

The array IBOUND indicates the status of every cell in the grid with the following codes.

Code	Status
zero	inactive
positive	variable head
negative	constant head

The values in the IBOUND array are read by the BAS1RP module; transmissive properties are read by module BCF1RP. This module (SBCF1N) insures that all transmissive parameters are equal to zero for cells designated inactive by the IBOUND array and that cells are designated "inactive" if all transmissive parameters are equal to zero.

Module SBCF1N is called by module BCF1RP and calls submodule SBCF1C. The SBCF1N module performs these functions in the following order:

1. Check the cell to see if it is designated inactive (IBOUND = 0). If it is inactive, set the vertical leakance (temporarily stored in CV), transmissivity (temporarily stored in CC), and hydraulic conductivity equal to zero.

2. Check the cell that is designated active to insure that there is at least one nonzero transmissive parameter. If there are no such nonzero transmissive parameters, designate the cell inactive and print an error message.

   (a) If the transmissivity is constant (LAYCON = 0 or 2), the transmissivity or vertical-hydraulic conductivity must be nonzero.

   (b) If the transmissivity is a function of head (LAYCON = 1 or 3), the hydraulic conductivity or vertical conductance must be nonzero.

3. Calculate the horizontal-branch conductances for layers where the transmissivity is constant (LAYCON = 0 or 2). Submodule SBCF1C is invoked to calculate the branch conductance from the transmissivity and cell dimensions.

4. Multiply the vertical leakance between cells (temporarily stored in CV) by the cell dimensions to get the vertical conductance.

5. If the simulation is transient, multiply the primary storage coefficient by DELR and DELC to get the primary storage capacity (SC1).

6. If the layer is confined/unconfined, multiply the secondary storage coefficient by DELR and DELC to get the secondary storage capacity (SC2).

7. RETURN.

Flow Chart for Module SBCF1N

LAYER TYPES are designated in the LAYCON table. Layer types are:

0 - confined
1 - unconfined
2 - constant/unconfined but transmissivity is constant
3 - confined/unconfined

Primary storage capacity is taken as specific yield times cell area for unconfined layers, and as confined storage coefficient times cell area for confined or confined/unconfined layers.

Secondary storage capacity is defined for confined/unconfined aquifers and is always taken as specific yield times cell area.



```
      SUBROUTINE SBCF1N(HNEW,IBOUND,SC1,SC2,CR,CC,CV,HY,TRPY,DELR,DELC,
     1          ISS,NCOL,NROW,NLAY,IOUT)
C
C-----VERSION 1642 15MAY1987 SBCF1N
C
C     ******************************************************************
C     INITIALIZE AND CHECK BCF DATA
C     ******************************************************************
C
C        SPECIFICATIONS:
C     ------------------------------------------------------------------
C
      DOUBLE PRECISION HNEW,HCNV
C
      DIMENSION HNEW(NCOL,NROW,NLAY),IBOUND(NCOL,NROW,NLAY)
     1    ,SC1(NCOL,NROW,NLAY),CR(NCOL,NROW,NLAY)
     2    ,CC(NCOL,NROW,NLAY),CV(NCOL,NROW,NLAY)
     3    ,HY(NCOL,NROW,NLAY),TRPY(NLAY),DELR(NCOL),DELC(NROW)
     4    ,SC2(NCOL,NROW,NLAY)
C
      COMMON /FLWCOM/LAYCON(80)
C     ------------------------------------------------------------------
C
C1------IF IBOUND=0, SET CV=0., CC=0., AND HY=0.
      KB=0
      DO 30 K=1,NLAY
      IF(LAYCON(K).EQ.3 .OR. LAYCON(K).EQ.1) KB=KB+1
      DO 30 I=1,NROW
      DO 30 J=1,NCOL
      IF(IBOUND(J,I,K).NE.0) GO TO 30
      IF(K.NE.NLAY) CV(J,I,K)=0.
      IF(K.NE.1) CV(J,I,K-1)=0.
      CC(J,I,K)=0.
      IF(LAYCON(K).EQ.3 .OR. LAYCON(K).EQ.1) HY(J,I,KB)=0.
   30 CONTINUE
C
C2------INSURE THAT EACH ACTIVE CELL HAS AT LEAST ONE NON-ZERO
C2------TRANSMISSIVE PARAMETER. IF NOT, CONVERT CELL TO NOFLOW.
      HCNV=888.88
      KB=0
      DO 60 K=1,NLAY
      IF(LAYCON(K).EQ.1 .OR. LAYCON(K).EQ.3) GO TO 55
C2A-----WHEN LAYER TYPE 0 OR 2, TRANSMISSIVITY OR CV MUST BE NONZERO
      DO 54 I=1,NROW
      DO 54 J=1,NCOL
      IF(IBOUND(J,I,K).EQ.0) GO TO 54
      IF(CC(J,I,K).NE.0.) GO TO 54
      IF(K.EQ.NLAY) GO TO 51
      IF(CV(J,I,K).NE.0.) GO TO 54
   51 IF(K.EQ.1) GO TO 53
      IF(CV(J,I,K-1).NE.0.) GO TO 54
   53 IBOUND(J,I,K)=0
      HNEW(J,I,K)=HCNV
      WRITE(IOUT,52) K,I,J
   52 FORMAT(1X,'NODE (LAYER,ROW,COL)',3I4,
     1      ' ELIMINATED BECAUSE ALL CONDUCTANCES TO NODE ARE 0')
   54 CONTINUE
      GO TO 60
```

```
C
C2B-----WHEN LAYER TYPE IS 1 OR 3, HY OR CV MUST BE NONZERO
   55 KB=KB+1
      DO 59 I=1,NROW
      DO 59 J=1,NCOL
      IF(IBOUND(J,I,K).EQ.0) GO TO 59
      IF(HY(J,I,KB).NE.0.) GO TO 59
      IF(K.EQ.NLAY) GO TO 56
      IF(CV(J,I,K).NE.0.) GO TO 59
   56 IF(K.EQ.1) GO TO 57
      IF(CV(J,I,K-1).NE.0.) GO TO 59
   57 IBOUND(J,I,K)=0
      HNEW(J,I,K)=HCNV
      CC(J,I,K)=0.
      WRITE(IOUT,52) K,I,J
   59 CONTINUE
   60 CONTINUE
C
C3------CALCULATE HOR. CONDUCTANCE(CR AND CC) FOR CONSTANT T LAYERS
      DO 65 K=1,NLAY
      KK=K
      IF(LAYCON(K).EQ.3 .OR. LAYCON(K).EQ.1) GO TO 65
      CALL SBCF1C(CR,CC,TRPY,DELR,DELC,KK,NCOL,NROW,NLAY)
   65 CONTINUE
C
C4------MULTIPLY VERTICAL LEAKANCE BY AREA TO MAKE CONDUCTANCE
      IF(NLAY.EQ.1) GO TO 69
      K1=NLAY-1
      DO 68 K=1,K1
      DO 68 I=1,NROW
      DO 68 J=1,NCOL
      CV(J,I,K)=CV(J,I,K)*DELR(J)*DELC(I)
   68 CONTINUE
C
C5------IF TRANSIENT MULTIPLY PRIMARY STORAGE COEFFICIENT BY DELR &
C5------DELC TO GET PRIMARY STORAGE CAPACITY(SC1).
   69 IF(ISS.NE.0) GO TO 100
      KT=0
      DO 80 K=1,NLAY
      DO 70 I=1,NROW
      DO 70 J=1,NCOL
      SC1(J,I,K)=SC1(J,I,K)*DELR(J)*DELC(I)
   70 CONTINUE
C
C6------IF LAYER IS CONF/UNCONF MULTIPLY SECONDARY STORAGE COEFFICIENT
C6------BY DELR AND DELC TO GET SECONDARY STORAGE CAPACITY(SC2).
      IF(LAYCON(K).NE.3 .AND. LAYCON(K).NE.2) GO TO 80
      KT=KT+1
      DO 75 I=1,NROW
      DO 75 J=1,NCOL
      SC2(J,I,KT)=SC2(J,I,KT)*DELR(J)*DELC(I)
   75 CONTINUE
C
   80 CONTINUE
C
C7------RETURN
  100 RETURN
      END
```

List of Variables for Module SBCF1N

Variable	Range	Definition
CC	Global	DIMENSION (NCOL,NROW,NLAY), Conductance in the column direction. CC(J,I,K) contains conductance between nodes (J,I,K) and (J,I+1,K). This array is used to temporarily hold transmissivity.
CR	Global	DIMENSION (NCOL,NROW,NLAY), Conductance in the row direction. CR(J,I,K) contains conductance between nodes (J,I,K) AND (J+1,I,K).
CV	Global	DIMENSION (NCOL,NROW,NLAY-1), Conductance in the vertical direction. CV(J,I,K) contains conductance between nodes (J,I,K) and (J,I,K+1). This array is used to temporarily hold Vcont.
DELC	Global	DIMENSION (NROW), Cell dimension in the column direction. DELC(I) contains the width of row I.
DELR	Global	DIMENSION (NCOL), Cell dimension in the row direction. DELR(J) contains the width of column J.
HCNV	Module	Indicator in the HNEW array that the cell is inactive.
HNEW	Global	DIMENSION (NCOL,NROW,NLAY), Most recent estimate of head in each cell. HNEW changes at each iteration.
HY	Package	DIMENSION (NCOL,NROW,NBOT), Hydraulic conductivity of the cell. (NBOT is the number of layers where LAYCON = 1 or 3.)
I	Module	Index for rows.
IBOUND	Global	DIMENSION (NCOL,NROW,NLAY), Status of each cell. < 0, constant-head cell = 0, inactive cell > 0, variable-head cell
IOUT	Global	Primary unit number for all printed output. IOUT = 6.
ISS	Package	Flag. = 0, simulation is transient. ≠ 0, simulation is steady state.
J	Module	Index for columns.
K	Module	Index for layers.
KB	Module	Index for bottom of layers.
KK	Module	Temporary variable set equal to K. KK is used as an actual argument in subroutine calls to avoid using the DO loop variable K as an argument, which causes problems with some compilers.
KT	Module	Index for top of layers.
K1	Module	NLAY-1.
LAYCON	Package	DIMENSION(80), Layer type code: 0 - Layer strictly confined. 1 - Layer strictly unconfined. 2 - Layer confined/unconfined (transmissivity is constant). 3 - Layer confined/unconfined (transmissivity varies).
NCOL	Global	Number of columns in the grid.
NLAY	Global	Number of layers in the grid.
NROW	Global	Number of rows in the grid.
SC1	Package	DIMENSION (NCOL,NROW,NLAY), Primary storage capacity of each cell (S*DELC*DELR).
SC2	Package	DIMENSION (NCOL,NROW,NTOP), Secondary storage capacity of each cell in the grid. (NTOP is the number of layers for which LAYCON = 2 or 3.)
TRPY	Package	DIMENSION (NLAY), Ratio of transmissivity in the column direction to transmissivity in the row direction.